

Datorövning 1 med Maple

Under denna datorövning skall vi lösa uppgifter från övningshäftet med hjälp av Maple. Vi skall rita kurvor och ytor. Syftet är att man efter övningen skall känna igen de vanliga andragsgradsytor, lättare kunna se kurvor, ytor och områden i tre dimensioner. Dessutom skall man kunna använda Maple för att plotta grafer till funktioner som dyker upp i kursen. Det är *inte* viktigt att hinna med alla uppgifter. Det är viktigare att ta sig tid med att fundera över vad man ser, hur man kan förbättra det och svara på de frågor som ställs.

Förberedelser

Läs igenom stencilen 'Introduktion till Maple' *innan* datorövningen. Läs *noggrant* igenom avsnittet om ytor och kurvor i denna handledning.

Tag med:

Denna handledning, stencilen 'Introduktion till Maple', övningshäftet och läroboken.

Kurvor och ytor

Kurvor i planet

Plana kurvor kan framställas på olika sätt och ritas med olika kommandon.

Funktionskurvor, $y = f(x)$, $a \leq x \leq b$ ritas i Maple med kommandot

```
plot(f(x), x=a..b);
```

Vill man exempelvis rita kurvan $y = \sin x$, $-\pi \leq x \leq \pi$ skriver man

```
plot(sin(x), x=-Pi..Pi);
```

Axlarna skalas automatiskt så att bilden fyller grafikfönstret maximalt. Om man vill ha samma skala på båda axlarna kan man klicka på knappen märkt **1:1** på menyraden. Alternativt kan man gå in på menyraden vid *Plot* och välja *Projection Constrained*. Menyn kan även tas fram med höger musknapp. Man kan också göra olika val av axlar, linjetjocklek, linjestil mm från menyraden. Dessa val kan alternativt göras direkt från kommandoraden genom att lägga till en eller flera parametrar (*options*) efter grundkommandot. Det går också göra färgval, precisera antalet indelningspunkter (t. ex. `numpoints=1000`), ange begränsningar i y -led med mera. (Se vidare hjälpen till kommandot `plot`.) Exempelvis ger kommandot

```
plot(x-x^3/6, x=-5..5, y=-6..6, scaling=constrained, color=blue, thickness=3);
```

en tjock blå kurva med samma skala på bägge axlarna, där y -axeln begränsats till det givna intervallet.

För att rita flera kurvor i en figur använder man `[]`-parenteser (även `{ }`-parenteser fungerar). Exempelvis kan man rita $\sin x$ och tre polynom i det nedan givna området med kommandot

```
plot([sin(x), x, x-x^3/6, x-x^3/6+x^5/120], x=-5..5, y=-5..5);
```

Vad är det för speciellt med polynomen?

Implicit givna kurvor, (nivåkurvor), $f(x, y) = K$. Här måste man först ladda ner fler grafikrutiner med kommandot `with(plots)`; och kan sen rita med kommandot

```
implicitplot(f(x,y)=K, x=a..b, y=c..d);
```

Man kan testa sig fram till lämpliga intervall för variablerna x, y . Väljer man för stora intervall får man sämre bilder. Exempelvis kan man rita kurvan $\frac{x^2}{4} + \frac{y^2}{9} = 1$

```
implicitplot(x^2/4+y^2/9=1,x=-5..5,y=-5..5,scaling=constrained);
```

Genom att tänka efter eller testa kan man hitta bättre variabelintervall. Det går också bra att rita flera kurvor, med givna färgval med t. ex.

```
implicitplot([x^2/4+y^2/9=1,x^2/9+y^2/4=1],x=-3..3,y=-3..3,
             color=[red,blue], scaling=constrained);
```

Kurvor på parameterform, $(x, y) = (x(t), y(t))$, $a \leq t \leq b$ ritas med kommandot

```
plot([x(t),y(t),t=a..b]);
```

Observera: Parameterintervallet skrivs *innanför* hakparentesen.

Enhetscirkeln $(x, y) = (\cos t, \sin t)$, $-\pi \leq t \leq \pi$ ritas med

```
plot([cos(t),sin(t),t=-Pi..Pi],scaling=constrained);
```

Ibland är det bra att först definiera och lagra bilderna och sen titta på dem. Om man vill se enhetscirkeln och en sinuskurva i samma bild kan man göra så här: Låt A betyda bilden av enhetscirkeln och B bilden av sinuskurvan, skriv

```
A:=plot([cos(t),sin(t),t=0..2*Pi],color=red):
B:=plot(sin(x),x=0..2*Pi,color=green):
```

Avsluta här med `:`, ej med `;`. (Avslutar man med `;` här fylls skärmen av siffror.) Maple visar nu inga bilder utan lagrar bilderna A resp B . Om vi nu skriver

```
display([A,B]);
```

visar Maple båda kurvorna i en bild. Även här kan man lägga till parametrar, t. ex.

```
display([A,B],scaling=constrained,view=[-1..2*Pi,-1..1]);
```

Optionen `view` begränsar området i x - resp y -led. Skriver man bara `view=c..d` begränsas området i y -led. Observera att A och B nu har en fastlagd betydelse och behåller denna tills man nollställer dem, med `A:='A'`; `B:='B'`; , eller tilldelar dem en annan betydelse.

Andra koordinatsystem, polära koordinater

I Maple finns flera olika koordinatsystem inlagda, t. ex. polära koordinater. Vill man rita kurvor som givna i polär form lägger man till parametern `coords=polar`. Exempelvis får man Arkimedes spiral ($r = \theta$) med kommandot

```
plot(t,t=0..20,coords=polar,scaling=constrained);
```

Kurvor i rummet

Här har vi endast parameterformen $(x, y, z) = (x(t), y(t), z(t))$, $a \leq t \leq b$. Kurvan ritas med kommandot

```
spacecurve([x(t),y(t),z(t)],t=a..b);
```

Observera: Parameterintervallet skall skrivas *utanför* hakparentesen. Här skiljer sig alltså de tre- och tvådimensionella fallen åt. Exempelvis fås kurvan $(x, y, z) = (t \cos t, t \sin t, t)$, $0 \leq t \leq 3\pi$ med

```
spacecurve([t*cos(t),t*sin(t),t],t=0..3*Pi);
```

Ytor i rummet

Ytor i rummet kan framställas på olika sätt och ritas med olika kommandon.

Funktionsytor $z = f(x, y)$, $a \leq x \leq b$, $c \leq y \leq d$ ritas i Maple med kommandot

```
plot3d(f(x,y),x=a..b,y=c..d);
```

Exempelvis fås funktionsytan $z = xe^{-x^2-y^2}$, $-2 \leq x \leq 2$, $-2 \leq y \leq 2$ med

```
plot3d(x*exp(-x^2-y^2),x=-2..2,y=-2..2);
```

Man kan vrida och vända på bilden genom att placera musmarkören i grafikfönstret och hålla vänster musknapp nere och flytta på musen eller klicka pilarna bredvid vinkelbeteckningarna ϑ och ϕ i menyraden. Med hjälp av menyer kan man göra olika val för 'axes', 'color' och 'style'. Väljer man *Patch and contour* under *Style* visas kurvor på funktionsytan där $f(x, y) = \text{konstant}$. Nivåkurvorna är projektionen av dessa kurvor på xy -planet. Vill man titta på nivåkurvorna (tvådimensionell bild), skriv

```
contourplot(f(x,y),x=a..b,y=c..d);
```

Här kan man välja antalet nivåkurvor, med t. ex. med parametern `contours=10`, eller ange specifika nivåkurvor, med t. ex. `contours=[-2,-1,0,1,2]`; För att få en mer färgglad bild lägg till parametern `filled=true`. Vill man välja färger lägg till t. ex. `coloring=[blue,red]` (se vidare `?contourplot`).

Definitionsmängden behöver ej vara en rektangel, gränserna i y kan bero på x , pröva t. ex. med

```
plot3d(x*exp(-x^2-y^2),x=0..2,y=-x..x);
```

Vad är definitionsmängden här?

Implicit givna ytor (nivåytor), ritas i Maple med kommandot

```
implicitplot3d(F(x,y,z)=K, x=a..b,y=c..d,z=e..f);
```

Här kan man testa sig fram till lämpliga intervall för variablerna x, y, z . Väljer man för stora intervall får man sämre bilder. Exempelvis kan sfären $x^2 + y^2 + z^2 = 1$ ritas med

```
implicitplot3d(x^2+y^2+z^2=1,x=-1..1,y=-1..1,z=-1..1);
```

Ytor på parameterform $(x, y, z) = (x(s, t), y(s, t), z(s, t))$, $a \leq s \leq b$, $c \leq t \leq d$ ritas i Maple med kommandot

```
plot3d([x(s,t),y(s,t),z(s,t)],s=a..b,t=c..d);
```

Observera: Parameterintervallen här skrivs utanför hakparentesen.

Cylindern $y^2 + z^2 = 1$ längs x -axeln, $-2 \leq x \leq 2$, kan skrivas på parameterform som $(x, y, z) = (s, \cos t, \sin t)$, $-2 \leq s \leq 2$, $0 \leq t \leq 2\pi$ och ritas med kommandot

```
plot3d([s,cos(t),sin(t)],s=-2..2,t=0..2*Pi,scaling=constrained);
```

Andra koordinatsystem, cylindriska och sfäriska koordinater

I Maple finns flera olika koordinatsystem inlagda, t. ex. cylindriska och sfäriska koordinater. Cylindriska koordinater använder variablerna r, t, z , där r och t är planpolära koordinater i xy -planet ($x = r \cos t$, $y = r \sin t$). $r = \sqrt{x^2 + y^2}$ är avståndet till z -axeln. Då man ritas med kommandot `cylinderplot` anger man r som funktion av vinkeln t och höjden z man skriver

```
cylinderplot(r(t,z),t=a..b,z=c..d);
```

eller helt ekvivalent

```
plot3d(r(t,z),t=a..b,z=c..d,coords=cylindrical);
```

Detta är speciellt lämpat för att rita cylindrar, och andra ytor med rotationssymmetri kring z -axeln, snyggt. Vill man exempelvis rita paraboloiden $z = x^2 + y^2$ startar man med att lösa ut $r = \sqrt{x^2 + y^2}$ som funktion av z . Man får $r = \sqrt{z}$. Välj sen lämpliga intervall för vinkeln t och höjden z , tag t. ex.

```
cylinderplot(sqrt(z),t=0..2*Pi,z=0..2);
```

För att rita en cirkulär cylinder med radie 1 längs z -axeln med $-2 \leq z \leq 2$ räcker det att skriva

```
cylinderplot(1,t=0..2*Pi,z=-2..2);
```

För mera komplicerade ytor se ?cylinderplot.

Sfäriska koordinater är de samma som beskrivs på sid 51-52 i flerdimboken, $(x, y, z) = (r \sin \theta \cos \varphi, r \sin \theta \sin \varphi, r \cos \theta)$. Här är $r = \sqrt{x^2 + y^2 + z^2}$ avståndet från origo. Vid användning av kommandot `sphereplot` anger man r som funktion av de båda vinklarna och skriver

```
sphereplot(r(phi,theta),phi=a..b,theta=c..d);
```

En sfär med radie $r = 1$ kan ritas med

```
sphereplot(1,phi=0..2*Pi,theta=0..Pi,scaling=constrained);
```

(se vidare ?sphereplot och ?coords).

Uppgifter:

Starta Maple på **Linux**-datorer på följande sätt:

- Klicka på **Applications** i övre vänstra hörnet på skärmen.
- I fönstret, som öppnas, Välj **Math** och sedan **Maple Classic Worksheet**. (Välj *inte* enbart **Maple!**)

Börja med att ladda in fler grafikrutiner med

```
with(plots);
```

Kurvor i planet

1. Rita kurvorna i övning **2.1abc** i övningshäftet var för sig (t. ex. `plot(x^2, x=-3..3);`). Välj samma skala på bägge axlarna. Man behöver ej skriva om kommandona flera gånger utan kan gå tillbaka på skärmen och ändra i gamla kommandon. Rita alla tre i en bild. Undersök vilka ändringar man kan göra under menyerna *Style*, *Axes*, *Projection*. Använd parameterform för att rita kurvan i **2.1d**.

Ange en parameterform: _____

Rita sen alla fyra kurvorna i en bild. Det kan då vara bra att starta med att definiera och lagra kurvorna i abc och d separat (obs avsluta då med : ej med ;) och sen rita dem med `display`. Behöver man begränsa området i *y*-led, lägg till parametern `view=c..d` med lämpligt val av *c* och *d*.

2. Rita kurvorna i övning **2.2** med `implicitplot`. Välj lämpliga intervall för *x* och *y*. Man får sämre bilder om intervallen väljs alltför stora. För att lättare jämföra kurvorna, titta på dem i samma bild med specificerade färgval.

Vad är det för slags kurvor? Svar: _____

Hur skiljer sig kurvorna i a) och b) åt? _____

Hur skiljer sig kurvorna i a) och d) åt? _____

3. Rita kurvorna $\frac{x^2}{4} - \frac{y^2}{9} = 1$, $\frac{x^2}{4} - \frac{y^2}{9} = -1$ och $\frac{x^2}{4} - \frac{y^2}{9} = 0$ i samma bild (`implicitplot`). Ta inte för små intervall för *x* och *y*. För att få bättre bilder kan man öka antalet indelningspunkter genom att t. ex. lägga till parametern `numpoints=2000`. Hur påverkar talen 4 och 9 kurvorna i figuren? Vad har de tre kurvorna gemensamt?

Svar: _____

Rita $\frac{x^2}{4} - \frac{y^2}{9} = 1$ och $\frac{x^2}{4} - y^2 = 1$ i samma bild.

Skillnader och likheter? _____

Rita $\frac{x^2}{4} - \frac{y^2}{9} = -1$ och $\frac{x^2}{4} - y^2 = -1$ i samma bild.

Skillnader och likheter? _____

Pröva gärna några andra värden på nämnare och högerled.

4. Rita kurvorna i övning **3.16** i samma bild. Använd samma skala på bägge axlarna. Vilken typ av kurvor ser det ut att vara? Var (och hur) ser de ut att skära varandra?

Svar: _____

I mån av tid rita några av mängderna i övning **2.5**, **2.6** Dessa mängder kan åskådliggöras med `implicitplot`, rita då randen ($|x|$ skrivs `abs(x)`).

Det går också bra att använda olikheter (t. ex. `abs(x)+abs(y)<1`.) Vill man ha områdena fyllda med färg så lägg till parametern `filled=true`. Gör man ett färgval, t. ex. med

```
coloring=[green,red]
```

så är det området med den första färgen som uppfyller olikheten.

För att få noggrannare kurvor öka antalet indelningspunkter med `numpoints` eller `grid`.

Funktionsgrafer och nivåkurvor

5. Studera sadelytan

$$f(x, y) = x^2 - y^2$$

genom att rita grafen (`plot3d`). Lägg in lämpliga axlar och pröva de olika alternativen under menyerna *Axes*, *Style*, *Color*, *Projection*. Vrid och vänd på figurena. Titta på nivåkurvorna i en tvådimensionell bild (`contourplot`). Vilken typ av kurvor är nivåkurvorna till sadelytan?

Svar: _____

Ytan $z = y^3 - 3x^2y$ kallas apsadel. Rita grafen och fundera över namnet.

6. Undersök funktionerna

$$g(x, y) = x^2 + y^2 \quad \text{och} \quad h(x, y) = \sqrt{x^2 + y^2}$$

var för sig genom att rita funktionsgrafer och nivåkurvor. Vill man begränsa området i höjddled använd t. ex. `view=0..10`. (Här kan man använda `cylinderplot` om man vill ha snyggare grafer.) Vilken typ av ytor är funktionsgraferna? Vilken typ av kurvor är nivåkurvorna? Hur skiljer sig nivåkurvorna åt för g och h ?

Svar: _____

Vad händer med graf och nivåkurvor om man ändrar g till $g(x, y) = x^2 + 4y^2$?

Svar: _____

7. Undersök graf och nivåkurvor till $f(x, y) = y^2$. Hur ser graf och nivåkurvor ut?

Svar: _____

Andragsytor från boken sid 45

8. Rita hyperboloiderna

$$x^2 + y^2 - z^2 = 1 \quad \text{och} \quad x^2 + y^2 - z^2 = -1$$

var för sig. Vad är skillnaden mellan dem? Vilken yta fås i gränsfallet $x^2 + y^2 - z^2 = 0$?

Svar: _____

(Använd `implicitplot3d` eller om ni vill ha snyggare bilder `cylinderplot`. Lös då ut $r = \sqrt{x^2 + y^2}$ som funktion av z .)

9. Rita ytan $x^2 + y^2 = 1$ (tre dimensioner). Vad för slags yta är detta? _____
10. Sfären $x^2 + y^2 + z^2 = 1$ kan ritas med `implicitplot3d` eller `sphereplot`. Men här skall vi undersöka parameterframställningen (jfr boken sid 51–52)

$$(x, y, z) = (\sin \theta \cos \varphi, \sin \theta \sin \varphi, \cos \theta), \quad 0 \leq \theta \leq \pi, \quad 0 \leq \varphi \leq 2\pi .$$

Starta med att lagra en bild av hela sfären, t. ex. med

```
A:=plot3d([sin(s)*cos(t), sin(s)*sin(t), cos(s)], t=0..2*Pi, s=0..Pi):
```

(Här har vinklarna kallats s och t för att det skall bli lite mindre att skriva.)

Lagra sen en bild, med specificerat färgval, av en del av sfären genom att minska på vinkelintervallen. Titta sen på bilderna samtidigt med `display`. Välj samma skala på alla axlar. Testa olika val av vinkelintervall, detta för att se parametrarnas betydelse.

Hur skall man välja vinkelintervallen för att få övre halvan av sfären? _____

Hur skall man välja vinkelintervallen för att få den del av sfären där $x \geq 0, y \geq 0, z \geq 0$?

Svar: _____

Vilken del av sfären fås om man väljer $0 \leq \theta \leq \pi$, $0 \leq \varphi \leq \pi/6$?

Svar: _____

Vilken del av sfären fås om man väljer $\pi/6 \leq \theta \leq \pi/4$, $0 \leq \varphi \leq 2\pi$?

Svar: _____

För att ändå tydligare se parametrarnas betydelse skall vi lägga in kurvor där den ena vinkeln är fixerad (parameterkurvor). Välj ett fixt värde på den ena parametern (t. ex. $s = \pi/6$) och låt den andra variera. Använd kommandot `spacecurve` och lagra en kurva. För att kurvan skall synas på sfären välj tjocka linjer och en avvikande färg, skriv t. ex.

```
C:=spacecurve([sin(Pi/6)*cos(t),sin(Pi/6)*sin(t),cos(Pi/6)],
              t=0..2*Pi, thickness=4,color=blue):
```

Titta sen på kurvan tillsammans med sfären (`display`). Rita också någon kurva med den andra parametern fixerad. Hur ser parameterkurvorna s =konstant resp t =konstant på sfären ut?

Svar: _____

Tips: Man kan minska skrivarbetet om man startar med att tilldela

```
x:=sin(s)*cos(t): y:=sin(s)*sin(t): z:=cos(s):
```

Sen kan man lagra bilden av sfären med

```
A:=plot3d([x,y,z],t=0..2*Pi,s=0..Pi):
```

och bilden av kurvan med

```
C:=spacecurve(subs(s=Pi/6,[x,y,z]),t=0..2*Pi,thickness=4,color=blue):
```

Här har s bara tillfälligtvis fått värdet $\pi/6$, medan x, y och z behåller dessa värden ända till dess man tilldelar dem ett annat värde eller nollställer dem.

Om man även låter r variera och ändrar x, y, z till

```
x:=r*sin(s)*cos(t): y:=r*sin(s)*sin(t): z:=r*cos(s):
```

kan man titta på ytor där $s =$ konstant och r, t varierar genom att t. ex. skriva

```
plot3d(subs(s=Pi/4,[x,y,z]),r=0..1,t=0..2*Pi,scaling=constrained);
```

Motsvarande kan sen göras med t =konstant. Hur ser ytorna s =konstant respektive t =konstant ut?

Svar: _____

11. Ändra i parameterframställningen av enhetssfären ovan och rita istället en ellipsoid med halvaxlarna 1, 2 respektive 3.

Några extraövningar på gränsvärden för dem som har tid över

12. Rita funktionerna i **3.17b,d** och **e** med `plot3d` t. ex. i området $-1 \leq x \leq 1$, $-1 \leq y \leq 1$. Ser det ut som om funktionerna har något gränsvärde då $(x, y) \rightarrow (0, 0)$? I så fall vad. Om inte ser det ut att finnas några vägar in mot $(0, 0)$ där funktionen antar olika värden nära $(0, 0)$?

Svar: _____

Se efter vad som händer om man byter ut termen xy i nämnaren på **3.17e** mot $2xy$ eller $1.99xy$. Här kan det vara bra att begränsa området i z -led med `view`.

Tänk på att en Maplegraf inte bevisar någonting. Titta på grafen till **3.17f**. Hur ser det ut nära $(0, 0)$ (med antal indelningspunkter enligt 'default')? Finns något gränsvärde. (Titta gärna i facit.)

13. Titta också på graferna funktionerna i övning **3.20bde** och **3.21**. Ser funktionerna ut att närma sig något speciellt värde då $x^2 + y^2$ växer?

Svar: _____

14. Titta närmare på funktionen i ex. 3.26 i **boken sid 86–87**,

$$f(x, y) = x^2y/(x^2 + y)^2, \quad (x, y) \neq (0, 0) .$$

Vad händer då $(x, y) \rightarrow (0, 0)$? _____